# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=python-pytest-subtests
pkgver=0.10.0
_commit=d47a5ced94ed7e69d990640ad867280154f1f535
pkgrel=3
pkgdesc='unittest subTest() support and subtests fixture'
arch=('any')
license=('MIT')
url='https://github.com/pytest-dev/pytest-subtests'
depends=('python-attrs' 'python-pytest')
makedepends=('git' 'python-setuptools-scm')
source=("git+https://github.com/pytest-dev/pytest-subtests.git#commit=$_commit")
sha512sums=('SKIP')

build() {
  cd pytest-subtests
  python setup.py build
}

check() {
  cd pytest-subtests
  python setup.py egg_info
  PYTHONPATH="$PWD" pytest
}

package() {
  cd pytest-subtests
  python setup.py install --root="$pkgdir" --optimize=1
  install -Dm644 LICENSE -t "$pkgdir"/usr/share/licenses/$pkgname/
}

# vim:set ts=2 sw=2 et:
